Problem:
Score : 600 points

Problem Statement

Given is a positive integer N.

We will choose an integer K between 2 and N (inclusive), then we will repeat the operation below until N becomes less than K.

Operation: if K divides N, replace N with N/K; otherwise, replace N with N-K.

In how many choices of K will N become 1 in the end?

Constraints

2 \leq N \leq 10^{12}

N is an integer.

Input

Input is given from Standard Input in the following format:

N

Output

Print the number of choices of K in which N becomes 1 in the end.

Sample Input 1

6

Sample Output 1

3

There are three choices of K in which N becomes 1 in the end: 2, 5, and 6.

In each of these choices, N will change as follows:

When K=2: 6 \to 3 \to 1

When K=5: 6 \to 1

When K=6: 6 \to 1

Sample Input 2

3141

Sample Output 2

13

Sample Input 3

314159265358

Sample Output 3

9

Solution in Common Lisp (SBCL 1.1.14) (judge verdict: Accepted)

;; F - Division and Subtraction

(defun divisible (n d) (zerop (mod n d)))

(defun divisors (n)
  "約数のリスト"
  (loop for d from 1 while (<= (* d d) n)
        when (divisible n d)
        append (let ((q (floor n d)))
                 (if (eql d q) (list d) (list d q)))))

(defun goodp (n k)
  "kがnに対して条件を満たす"
  (or (divisible (1- n) k) ; kはn-1の約数
      (and (divisible n k)
           (goodp (floor n k) k))))

(defun solve (n)
  "nに対して条件を満たすkの個数"
  ; n-1の2以上の約数はnに対して条件を満たす.
  ; nの約数kのうちn/kに対して条件を満たすkはnに対しても条件を満たす.
  (+ (loop for k in (divisors n)
           count (and (not (divisible (1- n) k)) ; n-1の約数は別途数える
                      (goodp n k)))
     (1- (length (divisors (1- n)))))) ; 1以外のn-1の約数の個数

(defun main ()
  (let ((n (read)))
    (princ (solve n))))

(main)